
Abstract

To extend the Lazarus IDE, it is not necessary to edit the Lazarus sources. The
Lazarus IDE is extensible with custom packages, and offers an API to integrate a pack-
age in the IDE. In this article, we’ll examine one of the ways in which the Lazarus IDE
can be extended, namely by adding items to the "New" dialog.

1 Introduction

The Lazarus IDE is open source, so in principle, anyone can take the IDE sources, and add
custom extensions to the IDE. This is cumbersome, and is hard to maintain. Fortunately,
the Lazarus IDE interface offers a way to extend the functionality of the IDE. All that needs
to be done is to create a package which is compiled and installed in the IDE, after which
the functionality offered by the package will be available. This is no different from the way
Delphi does it.

To demonstrate this, a package will be made which hooks into the File-New dialog in the

IDE. It allows to create new projects, based on project templates, located in some directory.
On the basis of the files in the template directory, a new project is created, and filled with

files. The templates can contain variables, which will be substituted with values that the

user can supply, such as the project name, a name for the main form etc..

2 Template organisation

The template repository is a simple directory, with a subdirectory for each template. The
subdirectory may contain other directories, which will be copied as part of the project
creation process. In the directory for each template, 2 files should exist:

project.ini contains the name of the project, plus definitions of the variables used in the
template.

description.txt a file containing a more verbose description of the project. This descrip-
tion (a simple text) will be shown in the IDE when the project is selected.

The project.ini file is in windows .INI format. It contains 2 sections. The first section is
calledGlobal and can contain the following key=value pairs:

Name The name of the project.

Author The author of the project.



Description a one-line description of the project.

Recurse a boolean (0 or 1) which indicates whether subdirectories are part of the templare
or not.

Exclude a comma separated list of extensions to be excluded from variable substitution
(for instance binary files). These files will simply be copied.

The second section is call&thriables  and should contain key=Value pair for each
variable that should be substituted when creating the project files. The value should be a
one-line description, which will be shown in the IDE.

The following is a sampleroject.ini file:

[project]
Name=Console Application
Author=Michael Van Canneyt

[variables]

ProjName=Name of the project file.

MyMainFormName=Name for the main form of the application.
AppName=Name of the application class.

The ProjName variable does not need to be defined here. If it is encountered, it will
automatically be substituted with the project name.

Variables should be specified in a format similar to the GNU Makefile variables:
T$(MyMainFormName) = Class(TForm)
éﬁd;

HereMyMainFormName is the name of the variable.

Note that not only the sources may contain variables. The filenames themselves can also
contain variables, they will be substituted as well.

The IDE package will be splitin 2 main parts:

1. A class to manage templates, and to create projects or files from templates. It takes
care of copying files, and performs the necessary variable name substitution.

2. A class which registers the templates in the IDE.

The first class is implemented as a descendeftGillection , the second class is a
TProjectDescriptor decsendent. Th&ProjectDescriptor is a class, defined
by the IDE interface.

3 Managing the project templates

To manage the templates, throjectTemplates class is defined. It is a descendent
of TCollection with a quite simple interface:

Constructor Create(Const ATemplateDir : String);
Procedure Initialize(Const ATemplateDir : String);
Procedure CreateProject(Const ProjectName, ProjectDir: String;



Variables : TStrings);
Function IndexOfProject(Const ProjectName : String) : Integer;
Function ProjectTemplateByName(
Const ProjectName : String) : TProjectTemplate;
Property TemplateDir : String;
Property Names [Index : Integer] : String;
Property Templates[index : Integer] : TProjectTemplate;default;

The meaning of these methods should be obvious from their names:

Create creates a new template collection, and initializes it ffbemplateDir

Initialize Initializes the templates from directoATemplateDir . This can be used to
change the template directory.

IndexOfProject returns the index of a template, based on it's name. returns -1 if the
project was not found.

ProjectTemplateByName returns thél ProjectTemplate instance based on the name.

CreateProject will create a project, based on the template with ndramplateName .
The project will be created in directo®rojectDir  , and any variables will be
substituted from th&lame=Value pairs in theVariables  stringlist.

Names is an indexed property containing the names of the found templates.

Templates is the default property of thEProjectTemplates class, providing indexed
access to the individudlProjectTemplate instances.

The interesting methods in this class &rtialize andCreateProject . The ini-
tialize method initializes the template definitions. It simply looks for the subdirectories
in the given template directory, and initializeJ BrojectTemplate instance from the

found subdirectories:
procedure Initialize(const ATemplateDir: String);

Var
Info : TSearchRec;
D : String;

begin
Clear;
FTemplateDir:=IncludeTrailingPathDelimiter(ATemplateDir);
D:=FTemplateDir;
If FindFirst(D+*' faDirectory,Info)=0 then
try
Repeat
If ((Info.Attr and faDirectory)<>0)
and not ((Info.Name="") or (Info.Name="..")) then
With Add as TProjectTemplate do
InitFromDir(D+Info.Name);
Until FindNext(Info)<>0;
finally
FindClose(Info);
end;
end;



This method could be expanded with code to ignore certain directories, for insEts&e
or.svn directories.

TheCreateProject method passes the actual work to the neéidemjectTemplate
instance:

procedure CreateProject(const ProjectName, ProjectDir; String;
Variables : Tstrings);

Var
T : TProjectTemplate;

begin
T:=ProjectTemplateByName(ProjectName);
T.CreateProject(ProjectDir,Variables);

end;
The CreateProject method ofTProjectTemplate is presented below.
The TProjectTemplate class is a descendent dCollectionltem , and has the

following properties:

Property Name : String;

Property Directory : String;

Property Description : String;

Property Variables : TStrings;
Property Author : String;

Property Recurse : Boolean;

Property Exclude : String;

Property FileCount : Integer;

Property Files[Index : Integer] : String;

The properties correspond to the values found irptiogect.ini file, except fofFileCount
which is the number of files found in the template, andites property, which gives the
names of all files found in the template. THariables contain the variables as defined

in the project file, together with their descriptions, not the values that will be substituted.

The following public methods exist:

Procedure CreateProject(Const ProjectDir : String; Values : TStrings);
Procedure CreateFile(Fileindex : Integer; Source,Values : TStrings);
Procedure CreateFile(Const FileName: String; Source,Values : TStrings);
Procedure CreateProjectDirs(Const BaseDir : String; Values : TStrings);
Function TargetFileName(FN : String; Values : TStrings) : String;
Function TargetFileName(l : Integer; Values : TStrings) : String;

The meaning of these methods should be obvious:

CreateProject creates a project based on the template files, in dire®oojectDir
TheValues stringlist contains the values to be used when substituting variable.

CreateFile Creates a file from source file specified by index or flename. The contents of
the file, with variables expanded, is returnediource . The values for substitution
are taken fronVvalues .

CreateProjectDirs recreates, und&aseDir , the directory tree found in the project tem-
plate directory, expanding variable names with the values foulthines .



TargetFileName returs a filename, relative to the project directory, for a source filename
specified by index or name. It replaces any variables found in the filename with
values found in th&alues stringlist.

ThelnitFromDir method a private method is used to initialize the template. It reads the
template settings from thgroject.ini file, and retrieves the list of files in the project:

procedure TProjectTemplate.InitFromDir(const DirName: String);

Var
L : TStringList;
FN : String;
begin

FDirectory:=IncludeTrailingPathDelimiter(DirName);
L:=TStringList.Create;
Try
FN:=FDirectory+'project.ini’;
If FileExists(FN) then
begin
With TMemlnifile.Create(FN) do
try
FName:=ReadString(SProject,KeyName,DirName);
FAuthor:=ReadString(SProject,KeyAuthor,”);
FDescription:=ReadString(SProject,KeyDescription,”);
FRecurse:=ReadBool(SProject,KeyRecurse,False);
FExclude:=ReadString(SProject,KeyExclude,”);
If (FExclude<>") then
FExclude:=FExclude+',’;
ReadSectionValues(SVariables,FVariables);
Finally
Free;
end;
end;
FN:=Directory+'description.txt’;
If FileExists(FN) then
begin
L.LoadFromFile(FN);
FDescription:=L.Text;
end;
GetFileList(FDirectory);
Finally
L.Free;
end;
end;

The main method i€reateProject

procedure CreateProject(const ProjectDir: String;
Values: TStrings);

begin
CopyAndSubstituteDir(Directory,ProjectDir,Values);
end;



It leaves the real work t&€opyAndSubstitureDir , Which is a recursive method. Itfirst
copies all files in the the source directory to the target directory, expanding variables as it
goes. Then it calls itself for each subdirectory found in the source directory (if recursion
was not disabled):

procedure CopyAndSubstituteDir(Const SrcDir,DestDir :String;
Values: Tstrings);

Var
D1,D2 : String;
Info : TSearchRec;

begin
D1:=IncludeTrailingPathDelimiter(SrcDir);
D2:=IncludeTrailingPathDelimiter(DestDir);
If not ForceDirectories(D2) then
Raise ETemplateError.CreateFmt(SErrCouldNotCreateDir,[D2]);
If FindFirst(D1+'*',0,Info)=0 then
try
repeat
if (info.name<>'description.txt’)
and (info.name<>'project.ini’) then
CopyAndSubstituteFile(D1+Info.Name,
D2+SubstituteString(Info.Name,
Values),
Values);
Until (FindNext(Info)<>0);
finally
FindClose(Info);
end;
if Recurse then
If (FindFirst(D1+*',0,Info)<>0) then
try
repeat
if ((Info.attr and faDirectory)<>0) and
(Info.Name<>'.") and (info.Name<>'..") then
CopyAndSubstituteDir(D1+Info.Name,
D2+SubstituteString(Info.Name,
Values),
Values);
until FindNext(Info)<>0;
finally
FindClose(Info);
end;
end;

Note that variables in the names of the target files or directories are substituted with the
value of the variables using tl&ubstituteString function.

Copying a file is done in th€opyAndSubstituteFile method:

procedure TProjectTemplate.CopyAndSubstituteFile(Const SrcFN,DestFN :
String; Values : Tstrings);

Var



L : TStrings;

begin
If pos(ExtractFileExt(SrcFN)+',’,Exclude)<>0 then
begin
If not SimpleFileCopy(SrcFN,DestFN) then
Raise ETemplateError.CreateFmt(SErrFailedToCopyFile,[SrcFN,DestFN]);
end
else
begin
L:=TstringList.Create;
try
CreateFile(SrcFN,L,Values);
L.SaveToFile(DestFN);
Finally
L.Free;
end;
end;
end;

This method checks whether the file must be copied as-is (which is done in SimpleFile-
Copy), or whether the contents of the file must be checked for variables, in which case the
public CreateFile  method is used. This method is quite easy:

procedure TProjectTemplate.CreateFile(const FileName: String; Source,
Values: TStrings);

Var
F . Text;
Line : String;

begin
AssignFile(F,FileName);
Reset(F);
Try
While not EOF(F) do
begin
ReadLn(F,Line);
Source.Add(SubstituteString(Line,Values));
end;
Finally
CloseFile(F);
end;
end;

This covers the most important methods in fferojectTemplate class. The two
classes described here are implemented iptbgcttemplates unit. This unit is indepen-

dent of the Lazarus IDE, and could be used in any project. In the next section, we explain
how to use this unit in the Lazarus IDE.



4 The various Lazarus IDE interfaces

In Lazarus, the equivalent of the Delphi 'Open Tools API’ is the 'Lazarus IDE interface’.
This is a collection of units which expose various elements of the Lazarus IDE so they can
be used in packages. The interface consists of a set of base classes, together with some
global variables which are set by the IDE. These units are located idéhdf directory

in the Lazarus source directory. This directory contains the following units:

LazIDEIntf Contains a general interface to the IDE, which allows to open files, and re-
trieve some configuration information.

NewltemIntf Contains an interface to thiEile|New’ menu dialog. It allows to add item
categories, such as the pre-defined 'File’, 'Project’ or 'Package’.

Projectintf Contains an interface to create new items in the 'File’ and 'Project’ category
of the'File|New’ dialog. This is the unit which will be used below.

HelpIntf An interface to the help system of lazarus. The help system is very extensible.

ConfigStorage An interface to the lazarus configuration system. This can be used if an
IDE package needs to save/restore settings. The settings will be stored in the Lazarus
settings directory.

FormEditingIntf Defines interfaces to form and component editors.

Packagelntf An interface to the IDE package system. It can be used to introduce new
packages (in the 'Package’ category) in thiie|New’ dialog.

MacrolIntf An interface to the IDE Macros interface. The macros are used in the tools,
build commands, configuration settings. However, it does not include the possibility
to define additional macros, so it was not used to implement the project templates.

IDECommands An interface to IDE commands. This can be used to add command
keystrokes to the IDE.

ActionsEditor an interface to implement and register custbiiction descendents, which
works much as in Delphi.

ComponentEditor an interface to component editors, compatible to the Delphi version.
PropEdits an interface to property editors, compatible to the Delphi version.

Objectinspector Contains the Object Inspector class. It simply manages the various prop-
erty editors.

There are other units in this directory, with some examples of component editors.

5 The project interface

Of all the interfaces presented in the previous section, only the Lazarus IDE interface,
configuration storage, and the project interface will be used.

The project interface is defined in tReojectiIntf unit. It defines a number of classes which
can be used to extend the IDE with custom projects and files. The IDE itself uses the classes
in this interface to implement the standard classes.

The two most important classes in this unit are the following



TProjectDescriptor A class to implement a new project in the IDE. When the IDE needs
to create a new project, it does so based on the properties and methods found in this
class.

TProjectFileDescriptor A class to implement a new file (a unit or program file) in the
IDE. When the Lazarus IDE needs to create a new source file, it uses the methods
and properties if this class to do so.

How should these classes be used ? Simple: A descendent of these classes must be made,
and it should either set some properties, or override some methods so the Lazarus IDE can
use it to create a new project or file.

An instance of this descendent class must be registered in the Lazarus IDE, so the IDE is
aware of the new project type. When the IDE needs to create a new file of the registered
type, the methods of the instance will be used, as will be demonstrated below.

The first of the two classes that is needed to implement the template project$ R tjectDescriptor
class. It has the following (simplified) declaration:

TProjectDescriptor=class(TPersistent)

function DolnitDescriptor: TModalResult; virtual;

function GetlLocalizedName: string; virtual;

function GetlLocalizedDescription: string; virtual;

function InitProject(AProject: TLazProject): TModalResult;
virtual,

function CreateStartFiles(AProject: TLazProject): TModalResult;

virtual;

property Name: string;

property VisibleInNewDialog;

property Flags: TProjectFlags;

property DefaultExt;

end;

The Name property identifies the kind of project in the IDE, but is otherwise not used.
The VisibleInNewDialog property (standar@rue ) determines whether a project of
this type will be shown in the 'New’ dialog. If so, it will be shown with a caption as
returned byGetLocalizedName , and when itis selected, the description as returned by
GetLocalizedDescription will be shown.

When the user selects the kind of project as described by the descriptor, the following
happens:

1. IDE will call the DolnitDescriptor method. This method should be used to
show a dialog which can be used to ask some additional information of the user.

2. IftheDolnitDescriptor method returnsnrOK(the default behaviour), then the
IDE will discard the currently active project, and starts a new project.

3. After some internal initializatiornitProject is called with a new LazProject
instance, which represents the new project. TReojectDescriptor can set
some properties ofLazProject  (e.g. compiler options) or call its methods. If
this method returns a value other thanOK the new project is abandoned.

4. After thelnitProject has finished withmrOk, the CreateStartFiles rou-
tine is called. This method should also retannOK and should be used to add files
to the project.



After these methods are called, the new project is ready for use in the IDEedfeStartFiles
was used to create some files, then they can be opened in the IDE, or they may already be
open, depending on the options used.

TheTLazProject classthatis passed to the last two methods, has the following interface
declaration:

function CreateProjectFile(
const Filename: string): TLazProjectFile;

procedure AddFile(ProjectFile: TLazProjectFile;

AddToProjectUsesClause: boolean);
procedure RemoveUnit(Index: integer);
procedure AddSrcPath(const SrcPathAddition: string);
procedure AddPackageDependency(const PackageName: string);
property MainFilelD: Integer;
property Files[Index: integer]: TLazProjectFile;
property FileCount: integer read GetFileCount;
property MainFile: TLazProjectFile;
Property Title: String;
property Flags: TProjectFlags read FFlags write SetFlags;
property LazCompilerOptions: TLazCompilerOptions;
property ProjectinfoFile: string;

Of these CreateProjectFile must be called in order to create the progrén file.

The Project info file (thelpi) file name can be set using tReojectinfoFile property.
TheTitle property can be used to set the title of the application. Hites property
provides indexed access to all files defined within the project (as shown in the project
inspector), and in this indexed list, the file with indebainFileID is the main project

file (the program file). Thélags property contains a set of the following flags which
control the behaviour of the IDE, and how it treats the main source file for the project:

pfSaveClosedUnitsTells the IDE to save information about closed files (which are not
part of the project) in the project information file.

pfSaveOnlyProjectUnits Tells the IDE not to save information about foreign files

pfMainUnitlsPascalSource Tells the IDE that the main unit is a pascal source file even if
the extension is napas or .pp.

pfMainUnitHasUsesSectionForAllUnits Tells the IDE to add/remove all pascal units to
the main file's uses section.

pfMainUnitHasCreateFormStatements Tells the IDE to add/remov&pplication.CreateForm
statements to the main file for each form that must be auto-created.

pfMainUnitHasTitleStatement Tells the IDE to add/removefpplication.Title:=
statement to the main source file.

pfRunnable tells the IDE that this project can be run.

LazCompilerOptions is a reference to the compiler settings used when compiling this
project. The declaration of this class is very long, the reader is referred to the source of the
projectintf unit for details.

The CreateProjectFile must be used to create the main project file, as this is a
project file, not a unit. How the project file’s source is created depends oRldgs
property. Normal files/units can be added to the project withAtthéFile call. They can

be removed with thRemoveUnit path. The dependency of the project on other packages,
such as the LCL or FCL, can be indicated usingAldelPackageDependency call.

Armed with these classes, the template project descriptor can be declared as follows:

TTemplateProjectDescriptor = class(TProjectDescriptor)
Private

FTemplates : TProjectTemplates;

FTemplate : TProjectTemplate;

FProjectDirectory : String;

FProjectName : String;

FVariables : TStrings;
protected

Procedure InitTemplates;

procedure SaveTemplateSettings;
public

constructor Create;

destructor destroy;

Function DolnitDescriptor : TModalResult;

function GetLocalizedName: string;

function GetLocalizedDescription: string;

function InitProject(AProject: TLazProject) : TModalResult;

function CreateStartFiles(AProject: TLazProject) : TModalResult;
end;

TheFTemplates field will keep a reference to thEProjectTemplates defined ear-
lier. TheFTemplate variable will be used to keep the template chosen by the user, just
as theProjectDirectory , ProjectName andFVariables fields.

A single instance of this class is instantiated in Register  procedure of the package
in which the templates are implemented:

Var
TemplateProjectDescriptor : TTemplateProjectDescriptor;

procedure Register;

begin
TemplateProjectDescriptor:=TTemplateProjectDescriptor.Create;
RegisterProjectDescriptor(TemplateProjectDescriptor);

end;

TheRegisterProjectDescriptor call is defined in th@rojectintf unit. This means
that the project descriptor 'lives’ as long as the package is loaded, i.e., currently from the
start of the IDE till it is closed.

The functionsGetLocalizedName and GetLocalizedDescription determine
what the IDE shows to the user in the 'New’ dialog:

function TTemplateProjectDescriptor.GetLocalizedName: string;
begin

Result:="Template Project’;
end;

function TTemplateProjectDescriptor.GetLocalizedDescription: string;
begin

Result:='"Create a project, based on a series of project templates’;
end;

Figure 1: The entry in the 'New’ dialog
New..=m——m———————
File
1 - Unit

~Farm

+Datamodule

- custam program

- Tewt

- FPCUnit TestCase

- B Project

o Application

L Program

- Custom Pragram
=]

+-FPCUnit Console &pplication

- FPCUnit &pplication
[ TemplateProjec
Package

- standard Package

The result is shown in figufg 1 on pggg 12.
The constructor and destructor are quite straightforward:

constructor TTemplateProjectDescriptor.Create;
begin
inherited Create;
Name:='"TemplateProject’;
FVariables:=TStringList.Create;
end;

destructor TTemplateProjectDescriptor.destroy;
begin
FTemplate:=NlI;
FreeAndNil(FTemplates);
FreeAndNil(FVariables);
Inherited;
end;

Note that the=Templates is not initialized in the constructor. To save memory, initial-
ization is only done when the project descriptor is activated for the first time. Then, a call
to InitTemplates is made:

procedure TTemplateProjectDescriptor.InitTemplates;

Var
D,P : String;

begin

If (FTemplates=Nil) then
With GetIDEConfigStorage('projtemplate.xml’,True) do

try
P:=LazarusIDE.GetPrimaryConfigPath;
P:=IncludeTrailingPathDelimiter(P)+'templates’;
D:=GetValue('TemplateDir',P);
FTemplates:=TProjectTemplates.Create(D);

Finally
Free;

end;

end;

TheGetIDEConfigstorage callis part of theConfigStorage  interface to the IDE.

It manages all configuration files: it creates configuration files in a centralized location. The
return value of this function is an instanceTonfigStorage , which is much like an
ini-file or registry, only that the settings file has an XML format. Three parameter to

this call indicates that the stored info should be read from disk.Gét¥alue call is part

of theTConfigStorage  class:

function GetValue(APath, ADefault: String): String;
function GetValue(APath: String; ADefault: Integer): Integer;
function GetValue(APath: String; ADefault: Boolean): Boolean;

it reads a value (string, integer, boolean) from the XML file. The location in the XML file
is determined byAPath , which follows an XPATH-like syntax. In the code above, the
GetValue call is used to read the location of the template directory. Based on this value,
the templates are initialized.

The value ofP is used as a default value. It is used by #@onfigStorage  if no
appropriate value was found in the configuration file. The value isfobtained from the
Lazarus IDE interface: th&etPrimaryConfigPath call returns the location of the
Lazarus configuration files (more on this follows below). The templates are assumed to be
in a subdirectory of this location.

ThelnitTemplates procedure is called when the Lazarus IDE calldthigdescriptor ,
i.e. when the user has selected the project type corresponding to the project descriptor in
the 'New’ dialog:

function TTemplateProjectDescriptor.DolnitDescriptor: TModalResult;

begin

InitTemplates;

Result:=ShowOptionsDialog;

If (Result=mrOK) and (FVariables.Count<>0) then
Result:=ShowVariableDialog;

If (Result=mrOK) then
begin
FVariables.Values['ProjName’]:=FProjectName;
FVariables.Values['ProjDir']:=FProjectDirectory;
end;

After initializing the templates, the options dialog is shown: this allows the user to select
which template should be used, what the name of the project will be, and where the project
should be located, as can be seen in figure figure 2 or[page 14. If the user closed the dialog
succesfully, and there are variables for which a value must be asked, the variable dialog
is shown (figure figurg]|3 on pa@e|14). If the user closed this dialog with succes, then the
standard’rojName andProjDir  variables are defined.

The ShowOptionsDialog  function is quite simple, really:

Figure 2: The options dialog: selecting a template

function TTemplateProjectDescriptor.ShowOptionsDialog : TModalResult;

var
I: Integer;

begin
with TTemplateOptionsForm.Create(Application) do
try
Templates:=Self.FTemplates;
Result:=ShowModal;
if Result=mrOK then
begin
FProjectDirectory:=
IncludeTrailingPathDelimiter(ProjectDir);
FProjectName:=ProjectName;
FTemplate:=Template;
FVariables.Assign(FTemplate.Variables);
I:=FVariables.IndexOfName('ProjName’);
if (I<>-1) then FVariables.Delete(l);
I:=FVariables.IndexOfName('ProjDir’);
if (I<>-1) then FVariables.Delete(l);
end;
if SettingsChanged then
SaveTemplateSettings;
finally
Free;
end;
end;

If by any chance, th€rojName andProDir are in the list of variables, they are deleted,

so they will not be shown to the user when values are asked for other variables. If the
user pressed the configure button in the dialog (indicated bys#tengsChanged
property), then the new settings are saved inSheeTemplateSettings method:

procedure TTemplateProjectDescriptor.SaveTemplateSettings;

begin
With GetIDEConfigStorage('projtemplate.xml’,False) do
try
SetValue('TemplateDir’,FTemplates. TemplateDir);
WriteToDisk;
Finally
Free,
end;
end;

This method does essentially the opposite of lhiEremplates method discussed
above. Note that th&etIDEConfigStorage gets a second parameter which indicates
that it is only needed for writing all settings; any previous information in the file will be
lost.

After all this, thelnitDescriptor call of the project descriptor has finished. If it
returnedmrOK the IDE will now close any open project and start a new project. it will
then call thenitProject method:

function InitProject(AProject: TLazProject) : TModalResult;

Var
| : Integer;
AFile: TLazProjectFile;
FN : String;
B : Boolean;
RFN : String;
L : TStringList;

begin
AProject.AddPackageDependency('FCL);
AProject.AddPackageDependency('LCL");
AProject.Title:=FProjectName;
FTemplate.CreateProjectDirs(FProjectDirectory,FVariables);
AProject.ProjectinfoFile:=FProjectDirectory
+FProjectName+’.Ipi’;
For 1:=0 to FTemplate.FileCount-1 do
begin
FN:=FTemplate.FileNamesJl];
B:=CompareText(ExtractFileExt(FN),".Ipr’)=0;
If B then
begin
FN:=FProjectDirectory+
FTemplate.TargetFileName(FN,FVariables);
AFile:=AProject.CreateProjectFile(FN);
AFile.IsPartOfProject:=true;
AProject.AddFile(AFile,Not B);
AProject.MainFilelD:=0;
L:=TstringList.Create;
try
FTemplate.CreateFile(l,L,FVariables);
AFile.SetSourceText(L.Text);
Finally
L.Free;
end;
end;
end;
Result:=mrOK;
end;

This function starts by adding dependencies on the FCL and LCL packages to the project,
and sets the project title. Based on the settings supplied by the user, It then proceeds by
creating all directories in that exist in the template project at the new project location. It
sets the name of the project information file. It then searches for the project file in the
template, and creates a file desciptor for this viaGneateProjectFile method of
TLazProject . This returns arLazProjectFile instance, which is added to the
project as the main project file. Last but not least, the source for the project file is loaded
from the template, and set via tBetSourceText  method of thelLazProjectFile

instance.

At this point, the Lazarus IDE has started the project, and has added the project source
file to it. Now the rest of the project files must still be created. This is done in the
CreateStartFiles function, which is called next by the IDE:

Function CreateStartFiles(AProject: TLazProject) : TModalresult;

Const
IDeOpts=[nflsPartOfProject,nfOpeninEditor,nfCreateDefaultSrc];

Var
Descr : TProjectFileDesc;
| . Integer;
FN,FN2 : String;
B : Boolean;
begin
Descr:=TProjectFileDesc.Create(FTemplate,FVariables);
Try
For 1:=0 to FTemplate.FileCount-1 do
begin
FN:=FTemplate.FileNames]l];
B:=CompareText(ExtractFileExt(FN),".Ipr')<>0;
If B then
begin
B:=CompareText(ExtractFileExt(FN),".Ifm")<>0;
If B then
begin
FN2:=ChangeFileExt(FN,".Ifm’);
B:=FileExists(FN2);
FN:=FProjectDirectory+
FTemplate.TargetFileName(FN,FVariables);
If B then
Descr.ResourceClass:=TForm;
Descr.Findex:=I;
LazarusIDE.DoNewEditorFile(Descr,FN,”,IdeOpts)
end;
end;
end;
Finally
Descr.Free;
end;
Result:=mrOK;
end;
The method starts by creating an instance offtReojectFileDesc class. This class

is used by Lazarus to describe a new file for a project, and will be discussed in more detalil
below. Then the procedure loops over all files in the template; it skips the project file and
the form files. For each other file it saves the index inTlReojectFileDesc instance,

and sets th&®esourceClass field to TForm if it finds a form file next to the unit file.

This will tell lazarus what class it should create in order to display the 'form’ (a module, a
form etc.).

Then it calls the Lazarus IDE interface:
LazarusIDE.DoNewEditorFile(Descr,FN,”,1deOpts)
This tells the Lazarus IDE to create a new file, based on the source file desbrgstor ,

with filenameFN and empty source. The options passeltleOpts tell the IDE that the
file should be part of the project, that the file should be opened in the editor, and that it

Figure 4: The result of the project template

Main application form

unit frmMain;
{$mode obj fpc} {$H+} |
interface

uses
Clagses, SysUtils,

type
{ TForml }

THMainForm = class(;

EClose: TButton;
private

{ private declarations }
public

{ public declarations }
end;

var
MainForm: TMainForm;

implementation

initialization
{$1 frmMain.lrs}

should create a default source for the file.

The actual work of creating the source file is then handled byl BrejectFileDesc
class, discussed in the next section.

At this point, the Lazarus IDE has finished creating the new project. It will look as in figure

[4 on pagé 1.

6 The file interface

In the previous section, thieProjectFileDesc class was used to create the source files
for the project. This class is a descendent BfojectFileDescriptor . This class

is used by the IDE when it needs to create a new file; This can be a unit, a form, any kind of
file. The IDE uses this class internally also to create the file items found in the IDE 'New’
dialog. The class has the following public interface:

function GetLocalizedName: string; virtual,

function GetLocalizedDescription: string; virtual;

function GetResourceSource: string; virtual;

function CreateSource(const Filename, SourceName,
ResourceName: string): string; virtual,

property Name: string;

property DefaultFilename: string;

property DefaultFileExt: string;

property DefaultSourceName: string;

property DefaultResFileExt: string;

property DefaultResourceName: string;

property RequiredPackages: string;

property ResourceClass: TPersistentClass;

property IsComponent: boolean read FIsComponent;
property UseCreateFormStatements: boolean;
property VisibleInNewDialog: boolean

property IsPascalUnit: boolean;

property AddToProject: boolean;

TheVisibleInNewDialog , GetLocalizedName andGetLocalizedDescription
property and methods serve the same purpose as their counterparfERndfertDescriptor
class.

The CreateSource  method is called by the IDE to create the source for the file: The
result should be the contents of the (pascal unit) file. The parameters passed are the file-
name, source file name and resource name (if any) that the IDE has assigned to this file. A
descendent should override this method to return the text of the source for this file.

The GetResourceSource  method should be overridden by descendents to return the
contents of the initiallfm file (as text) that matches the source file. It will only be called
if the ResourceClass class pointer is natlil , i.e. when the IDE has decided that this

is a form or datamodule or any visual object that needs streaming.

The meaning of the various properties should be straightforward:
DefaultFilename a default filename for this kind of file (with extension).
DefaultFileExt a default extension for this kind of file.
DefaultSourceNamethe default unit name.

DefaultResFileExt default extension for resource fildr§).
DefaultResourceNamedefault name for the resourceédqrm).
RequiredPackagesrequired packages for this kind of unit.

ResourceClassthe class that the IDE will create when displaying the form for this file.
Only when this property is ndtiil  will the IDE decide that a form (or datamodule)
must be displayed when loading this file.

IsComponent Should be set tGrue if ResourceClass is aTComponent Descen-
dent.

UseCreateFormStatementsshould be set tdrue if the IDE can add this resource to the
Application.CreateForm statements.

IsPascalUnit should be set tdrue if this file is a pascal unit.

AddToProject should be set to true if a file of this kind should be added to the project
when it is created.

For the project templates, the descendent is quite simple. It needs to override only the
minimum of calls:

TProjectFileDesc=class(TProjectFileDescriptor)
constructor Create(ATemplate : TProjectTemplate;
Values: TStrings);
Function GetResourceSource : String;override;

function CreateSource(const Filename, SourceName,
ResourceName: string): string; override;
function GetLocalizedName: string; override;
function GetLocalizedDescription: string; override;
end;

TheGetLocalizedName  andGetLocalizedDescription return simply a string,
and so will not be shown here. The constructor only saves the variables which are passed
to it, and initializes the name:

constructor TProjectFileDesc.Create(ATemplate: TProjectTemplate;
Values : TStrings);

begin

Inherited Create;

FTemplate:=ATemplate;

FVariables:=Values;

Name:='"Regular File’;
end;

The routines which do the actual work are (eateSource routine:

function TProjectFileDesc.CreateSource(const Filename, SourceName,
ResourceName: string): string;

Var
L : Tstrings;

begin
L:=TstringList.Create;
try
FTemplate.CreateFile(FIndex,L,FVariables);
Result:=L.Text;
Finally
L.Free;
end;
end;

It simply passes the work on to th®rojectTemplate class. Th&etResourceSource
does something similar:

function TProjectFileDesc.GetResourceSource: String;

Var
L : Tstrings;
FN : String;

begin
Result:=";
If (ResourceClass<>Nil) then
begin
L:=TstringList.Create;
try
FN:=ChangeFileExt(FTemplate.FileNames[FIndex],".Ifm’);
FTemplate.CreateFile(FN,L,FVariables);

Result:=L.Text;
Finally
L.Free;
end;
end;
end;

Note that it checks wheth&esourceClass s set, which is only the case if a form file
exists. TheCreateStartFiles method discussed above, has taken care of that.

7 Conclusion

In the above, we have shown how to create a custom project in the Lazarus IDE. In doing
so, various interfaces of the IDE have been used: the project interface and the file interface.
However, the scope has been rather limited:

e Only a ready-made TForm descendent was created, and no other resources (e.g. a
TDatamodule) were created: the interface offers more possibilities than were used
here.

e What is more, the lazarus IDE interface offers other ways of obtaining the same
functionality for a package that implements project templates. This interface was
only touched upon, an needs to be explored more deeply.

e The possibility of creating a whole new category of new items was left untouched;
This could be used to display the various templates immediatly in the 'New’ dialog.

Exploring these other functionalities will be left to a future contribution, in which the

project template functionality will be rewritten. At the same time it will be shown that
the interface offers the possibility of creating custom resources (forms/data modules).

